DEPARTMENT OF MATHEMATICAL SCIENCES
COLLEGE OF SCIENCES
OLD DOMINION UNIVERSITY
NORFOLK, VA 23529

THEORETICAL STUDIES OF SOLAR LASERS AND CONVERTERS

By

John Heinbockel, Principal Investigator

Progress Report
For the period May 15 to December 31, 1987

Prepared for the
National Aeronautics and Space Administration
Langley Research Center
Hampton, Virginia 23665

Under
Research Grant NAG-1-757
Dr. Robert C. Costen, Technical Monitor
SSD-High Energy Science Branch

Submitted by the
Old Dominion University Research Foundation
P. O. Box 6369
Norfolk, Virginia 23508

January 1988

Progress Report Research Grant NAG-1-757
For the period May 15, 1987-Dec 31, 1987

Iodine Laser Geometry.

The geometry and setup for the $n-C_3F_7I$ Iodine laser is illustrated in the Figure 1. We are interested in the steady state laser operation with axial lasant flow. The following is a description of the mathematical modeling of this system.

Chemical Kinetics.

The chemical kinetics associated with this iodine laser has been disscused in the references 1 through 6. Let R denote the radical $n-C_3F_7$ the chemical kinetics can be summarized as follows:

Photodissociation of RI and I_2

$$RI + h\nu_1 \underset{\xi_1}{\xrightarrow{0.27\ \mu m}} R + I^*$$

$$I_2 + h\nu_2 \underset{\xi_2}{\xrightarrow{0.5\ \mu m}} I + I^*$$

Laser action (stimulated emission)

$$I^* + h\nu_3 \xrightarrow{1.32\mu m} I + 2h\nu_3$$

Quenching of I^* (reaction rates have units $[cm^3/sec]$)

$$I^* + RI \xrightarrow{Q_1} I + RI$$

$$I^* + I_2 \xrightarrow{Q_2} I + I_2$$

$$I^* + R \xrightarrow{Q_3} I + R$$

$$I^* + R_2 \xrightarrow{Q_4} I + R_2$$

$$I^* + I \xrightarrow{Q_5} I + I$$

RI generation and two body recombinations (reaction rates have units $[cm^3/sec]$)

$$R + I^* \xrightarrow{k_1} RI$$

$$R + I \xrightarrow{k_2} RI$$

$$R + R \xrightarrow{k_3} R_2$$

$$R + RI \xrightarrow{k_4} R_2 + I$$

$$R + I_2 \xrightarrow{k_5} RI + I$$

$$R + RI \xrightarrow{k_6} R_2 + I^*$$

$$RI + I^* \xrightarrow{k_7} I_2 + R$$

$$I + RI \xrightarrow{k_8} I_2 + R$$

I_2 formation (reaction rates have units $[cm^6/sec]$)

$$I^* + I + RI \xrightarrow{c_1} I_2 + RI$$
$$I + I + RI \xrightarrow{c_2} I_2 + RI$$
$$I^* + I + I_2 \xrightarrow{c_3} I_2 + I_2$$
$$I + I + I_2 \xrightarrow{c_4} I_2 + I_2$$
$$I + I + R_2 \xrightarrow{c_5} I_2 + R_2$$
$$I + I^* + R_2 \xrightarrow{c_6} I_2 + R_2$$

Wall reactions (reaction rates have units $[cm^3/sec]$)
Wall reactions have been proposed but have not as yet been implemented into the model. Two possible wall reactions are:

$$I + I + wall \xrightarrow{v_1} I_2 + wall$$
$$R_2 + I + wall \xrightarrow{v_2} R + RI + wall$$

Values of parameters.

The reaction rate coefficients in the above chemical reactions are assumed to have the bounds given in the Table 1. This Table also gives the final values used in the computer simulation.

Differential equations for chemical kinetics.

Using the notation $[A]$ to denote the concentration of species A in units of cm^{-3}, the differential equations defining the chemical kinetics for the iodine laser can be expressed by the following set of coupled nonlinear differential equations involving the concentrations

$[RI]$, $[R]$, $[R_2]$, $[I_2]$, $[I^*]$ and $[I]$:

$$\frac{d[RI]}{dt} = k_1[R][I^*] + k_2[R][I] + k_5[R][I_2] - k_7[I^*][RI] - k_4[R][RI]$$
$$- k_6[R][RI] - \xi_1[RI] + v_2[R_2][I] - k_8[I][RI]$$
$$\frac{d[R]}{dt} = \xi_1[RI] - k_1[R][I^*] - k_2[R][I] - 2k_3[R]^2 - k_4[RI][R]$$
$$- k_6[RI][R] - k_5[R][I_2] + v_2[R_2][I] + k_7[RI][I^*] + k_8[I][RI]$$
$$\frac{d[R_2]}{dt} = k_3[R]^2 + k_6[RI][R] + k_4[RI][R] - v_2[R_2][I]$$
$$\frac{d[I_2]}{dt} = c_1[RI][I^*][I] + c_2[RI][I]^2 + c_3[I_2][I^*][I] + c_4[I_2][I]^2$$
$$- \xi_2[I_2] + k_7[RI][I^*] - k_5[R][I_2] + v_1[I]^2$$
$$+ c_5[I]^2[R_2] + k_8[RI][I] + c_6[I][I^*][R_2]$$
$$\frac{d[I^*]}{dt} = Q_y\xi_1[RI] + \xi_2[I_2] - k_1[R][I^*] - Q_2[I_2][I^*]$$
$$- c\sigma\rho([I^*] - \frac{1}{2}[I]) + k_6[R][RI] - Q_3[R][I^*] - Q_4[R_2][I^*]$$
$$- Q_5[I^*][I] - k_7[RI][I^*] - c_6[R_2][I^*][I]$$
$$\frac{d[I]}{dt} = \xi_2[I_2] + Q_1[RI][I^*] + Q_2[I_2][I^*] - 2c_5[I]^2[R_2] - k_8[I][RI]$$
$$+ c\sigma\rho([I^*] - \frac{1}{2}[I]) - c_1[RI][I^*][I] - 2c_2[RI][I]^2 - c_3[I_2][I^*][I]$$
$$- 2c_4[I_2][I]^2 - k_2[R][I] + k_4[RI][R] + Q_3[I^*][R] + Q_4[I^*][R_2]$$
$$+ Q_5[I^*][I] + k_5[R][I_2] - v_2[R_2][I] - 2v_1[I]^2 - c_6[R_2][I^*][I]$$

In the above differential equations we neglect compressibility and use the material derivative

$$\frac{d[A]}{dt} = \frac{\partial[A]}{\partial t} + \frac{\partial[A]}{\partial z}\frac{dz}{dt} = \frac{\partial[A]}{\partial t} + \frac{\partial[A]}{\partial z}\omega$$

where $\frac{dz}{dt} = \omega$ is the flow rate in the axial direction. Note that the above system of nonlinear differential equations conserves the masses of the species involved in the reactions and at any point z we have the integrals

$$[RI] + [R] + 2[R_2] = constant$$
$$[RI] + 2[I_2] + [I^*] + [I] = constant$$

In the above equations describing the reaction kinetics Q_y is the quantum yield; ξ_1, ξ_2 are the photodissociation rates which are dependent upon the solar simulator concentration c_0; σ is the stimulated emission cross section; and ρ is the photon density in the optical cavity. As a first approximation we assume that

$$\xi_1 = (3.04(10^{-3}))c_0 \qquad \xi_2 = (3.38(10^{-2}))c_0$$

For the light flux density of lasing photons ρ we let $\rho = \rho_+ + \rho_-$ where $\rho_+ = \rho_+(z,t)$ denotes the photon density propagating in the positive z direction and $\rho_- = \rho_-(z,t)$ denotes the photon density propagating in the negative z direction. The differential equations for these photon densities are given by

$$\frac{1}{c}\frac{\partial \rho_+}{\partial t} + \frac{\partial \rho_+}{\partial z} = \sigma\rho_+([I^*] - \frac{1}{2}[I])$$
$$\frac{1}{c}\frac{\partial \rho_-}{\partial t} - \frac{\partial \rho_-}{\partial z} = \sigma\rho_-([I^*] - \frac{1}{2}[I])$$

where c is the speed of light in the medium. In the above equations $\sigma[I^*]$ is the amplification factor resulting from population of the upper lasing level of the active medium and $-\frac{1}{2}\sigma[I]\rho_+$ is the decrease in photon density due to population of the lower lasing level.

Initial and Boundary conditions.

We consider the steady state solution to the above differential equations which are then subject to the initial condtions:

$$[RI] = 3.5(10^{16})P_0 \quad \text{at} \quad z = 0 \quad P_0 \text{ is the pressure in torr}$$

with all of the other concentrations being zero at $z = 0$.

Boundary conditions must be imposed upon the ρ_+, ρ_- solutions due to the reflectivities from the mirrors and Brewster windows of the geometry in Figure 2. In this geometry R_1, R_2, w are given and R_{1eff}, R_{2eff}, t_m are calculated. In this study $R_{1eff} = 0.95848$ and $R_{2eff} = 0.81634$ denote the effective reflectivities of the Brewster window-mirror combination where $w = 0.98$ denotes the transmission coefficient for the Brewster windows. The transmission coefficient of the output mirror is then given by $t_m = w(1 - R_2)$ where R_2 is the reflectivity of the output mirror and is obtained from the relation

$$R_{2eff} = w^2 R_2$$

This gives a transmission coefficient of $t_m = 0.147$.

At the faces $z = 0$ and $z = L$ the differential equations for ρ_+ and ρ_- are subject to the boundary conditions:

$$\rho_+(0) = R_{1eff}\rho_-(0) \qquad \rho_-(L) = R_{2eff}\rho_+(L)$$

For the steady state , the solutions for ρ_+ and ρ_- must satisfy the relation

$$\rho_+(z)\rho_-(z) = K_0 = constant$$

for all values of z. This condition requires that at $z = 0$

$$\rho_+(0)\rho_-(0) = R_{1eff}[\rho_-(0)]^2 = \frac{1}{R_{1eff}}[\rho_+(0)]^2 = K_0$$

and at $z = L$ we must have

$$\rho_+(L)\rho_-(L) = \frac{1}{R_{2eff}}[\rho_-(L)]^2 = R_{2eff}[\rho_+(L)]^2 = K_0$$

These conditions require that

$$\rho_+(0) = \sqrt{K_0 R_{1eff}} \quad \text{and} \quad \rho_+(L) = \sqrt{\frac{K_0}{R_{2eff}}}$$

A shooting method is used to determine the initial conditions assigned to the steady state value of ρ_+. For fixed values of R_{1eff} and R_{2eff} we quess at an initial value for K_0 and integrate the system of steady state differential equations from $z = 0$ to $z = L$ and compare the theoretical value of $\rho_+(L)$ with the calcualted value. This iteration continues until this boundary condition is satisfied, in which case the power transmission is given by

$$P = \epsilon_\mu t_m c \rho_+(L) \quad (W/cm^2)$$

where ϵ_μ is the energy quantum of lasing photons.

Using the conversion factor that one solar constant equals 1.35 Kw/m^2 and given that 12.9 Kw corresponds to 300 amps then the input power is assumed to be linear and

$$\text{Input power} = I(amps)\frac{12.9\,Kw}{300\,amps}$$

The output power is then given by

$$\text{output power} = P_d A_b$$

where P_d is the power density and A_b is the cross sectional area of the laser beam. The power density varies with the beam diameter and the input power. For a beam diameter of $d\,(cm)$ we have

$$\frac{Input\ power}{\pi \ell d} = P_d$$

where $\ell\,(cm)$ is the length of the tube section being radiated by the solar simulater. Using these relations we obtained good agreement between the theoretical and actual output power from the laser.

Conclusion. The Figures 3 and 4 are obtained by holding R_{1eff} constant and letting R_{2eff} vary. Thus, if the numerical integration stopped at some point z along the tube we determined the value of R_{2eff} necessary to satisfy the boundary conditions. For this value of R_{2eff} the transmission coefficient t_m and resulting power density was calculated.

Also a sensisitivity analysis was done on the parameters occuring in the differential equations describing the chemical kinetics. The results of this sensitivity study indicated that a changing of the coefficients

$$k_1,\ c_2,\ Q_1,\ k_2,\ Q_2$$

produced the largest changes in the power output.

REFERENCES

1. John W. Wilson, S.Raja , Y.J.Shiu , Solar-Simulator-Pumped Atomic Iodine Laser Kinetics, NASA Technical Paper 2182, August 1983.
2. John W. Wilson, Yeunggil Lee, Willard R. Weaver, Donald H. Humes, Ja H. Lee, Threshold Kinetics of a Solar-Simulator-Pumped Iodine Laser, NASA Technical Paper 2241, February 1984.
3. L.V. Stock, J.W. Wilson, R.J. DeYoung, A Model for the Kinetics of a Solar-Pumped Long Path Laser Experiment, NASA Technical Memorandum 87668, May 1986.
4. G.Breederlow, E. Fill, K.J. White, The High Power Iodine Laser, Springer Verlag, N.Y., 1983.
5. J.S. Cohen , O.P. Judd , High Energy Optically Pumped Iodine Laser I. Kinetics in an optically thick medium. , J. Appl. Phys. 55(7), 1 April 1984.
6. E.V. Arkhipova, B.L. Borovich, A.K. Zapalskii, Accumulation of excited Iodine Atoms in Iodine Photodissociation Laser. Analysis of kinetic equations. Sov. J. Quantum Electron. , Vol 6, No.6, June 1976.

Reaction coefficient	Lower bound	Nominal value	Upper bound	Reference	Value used
Q_1	0.476E-16	2.0E-16	8.40E-16	(3)	0.476E-16
Q_2	0.730E-11	1.9E-11	4.94E-11	(3)(5)	0.730E-11
Q_3	1.230E-18	3.7E-18	11.1E-18	(5)	1.23E-18
Q_4	1.570E-16	4.7E-16	14.1E-16	(5)	1.57E-16
Q_5	0.530E-14	1.6E-14	4.80E-14	(5)	0.53E-14
k_1	0.9E-13	5.6E-13	34.7E-13	(3)	0.9E-13
k_2	0.657E-11	2.3E-11	8.05E-11	(3)	8.05E-11
k_3	0.65E-12	2.6E-12	10.4E-12	(3)	0.65E-12
k_4	1.0E-16	3.0E-16	9.0E-16	(3)(5)	1.0E-16
k_5	0.33E-11	1.0E-11	3.0E-11	(5)	3.0E-11
k_6	1.0E-17	3.2E-17	10.24E-17	(3)	1.0E-17
k_7	1.5E-19	3.0E-19	4.5E-19	(3)	1.5E-19
k_8	0.533E-23	1.0E-23	4.8E-23	(5)	0.0
c_1	1.0E-33	3.2E-33	10.2E-33	(3)	1.0E-33
c_2	1.6E-32	8.5E-32	45.0E-32	(3)	45.0E-32
c_3	4.44E-32	8.5E-32	14.4E-32	(3)	4.44E-32
c_4	2.92E-30	3.8E-30	4.94E-30	(3)	4.94E-30
c_5	3.6E-31	4.8E-31	6.0E-31		3.6E-31
c_6	1.35E-32	1.8E-32	2.25E-32		0.0
v_1	0.33E-12	1.0E-12	3.0E-12		0.0
v_2	0.33E-11	1.0E-11	3.0E-11		0.0

Table 1. Parameter Values For Chemical Kinetics



FIGURE 1

GEOMETRY FOR STEADY STATE LASER WITH AXIAL LASANT FLOW



FIGURE 2

MIRROR REFLECTIVITIES



FIGURE 3
CONTOUR PLOT OF POWER OUTPUT



FIGURE 4

POWER OUTPUT SURFACE

APPENDIX

COMPUTER PROGRAM FOR SIMULATION OF LASER POWER

```
      PROGRAM NLAS7 (INPUT,OUTPUT,TAPE5=INPUT,TAPE6=OUTPUT,TAPE8)
C
C     MAIN PROGRAM
C
C     THIS VERSION CONTAINS AN AUTOMATICALLY CHOSEN VARIABLE STEP
C     SIZE.  IT ALSO CONTAINS WALL EFFECT REACTIONS.
C
      COMMON /BLK2/ K1,K2,K3,K4,K5,K6,K7,K8,C1,C2,C3,C4,Q1,Q2,A,AD,V1,V2
     1   ,GG
      COMMON /BLK3/ B,B2,B3,C,A00,B00,EPSNU,OMEGA,C5,C6,Q3,Q4,Q5
      COMMON /BLK4/ CHSI10,CHSI20,ABAR0,Z1BAR,LC
      COMMON /BLK7/ ABC,C00,C0,OMEG1,P,R1,R2,TM
      COMMON /BLK8/ ZOL
      REAL LC,K1,K2,K3,K4,K5,K6,K7,K8
      WRITE (6,10)
   10 FORMAT (1X,' START OF PROGRAM   ')
C
C      DEFAULT VALUES
C
      ZOL = 5.70
      P = 40.
      LC = 6.0
      C00 = .5E25
      R1 = .9975
      R2 = .975
      A = 0.0
      TM = 1-R2
      OMEG1 = 5000.
      CON = 2.0E4
      NAMELIST /PARAM/ P,OMEG1,CON,C00,R1,R2,TM,LC,ZOL,A
   20 CONTINUE
      READ (5,PARAM)
      IF (EOF(5)) GO TO 30
      GO TO 50
   30 WRITE (6,40)
   40 FORMAT (1X,'END OF FILE ENCOUNTERED-STOP')
      STOP 1313
C
C
C       P=PRESSURE, TORR
C       OMEG1=FLOW RATE, CM/SEC
C       CON=PEAK CONCENTRATION
C       C00=INITIAL GUESS AT RHO-PLUS AT ZERO
C           WHICH IS SQUARE OF (C00*R1)
C       R1= REFLECTIVITY AT LEFT END
C       R2= REFLECTIVITY AT RIGHT END
C       TM= TRANSMISSION COEFFICIENT (OUTPUT MIRROR)
C      ZOL=POINT ALONG AXIS WHERE MAXIMUM ILLUMINATION OCCURS
C     IN THE CASE ILLUMINATION IS A BELL SHAPED CURVE
C      IN THE CASE OF A SQUARE WAVE, 2*ZOL IS CUT OFF POINT
C        THE POINT 2*ZOL IS WHERE ILLUMINATION CUTS OFF
C      LC=LENGTH OF CAVITY
C
C
```

```
   50 CONTINUE
      CMIN = 1.0E10
      CMAX = 1.0E30
      CO = CON
      C11 = CON
      OMEGA = OMEG1
      WRITE (6,60)
   60 FORMAT (///)
      WRITE (6,70) ZOL,CON,OMEGA,COO,R1,R2,P
   70 FORMAT (1X,T30,'ZOL = ',E15.7,T55,'CON = ',E15.7,T80,'OMEGA = ',
     1    E15.7,/,1X,T5,'COO = ',E15.7,T30,' R1 = ',F10.7,T55,' R2 = ',
     2    F10.7,T81,' P =',E15.7)
C
C     SET UP COEFFICIENTS IN DIFFERENTIAL EQUATIONS
C
      CALL COEFFS
C
C     INTEGRATE DIFFERENTIAL EQUATIONS FROM Z=0 TO Z=LC
C
      X1 = COO
      CALL INTEG
C
C        INTERVAL HALVING SCHEME
C
C       W1 AND W2 ARE WEIGHTS FOR INTERVAL HALVING SCHEME FOR
C       DETERMINING COO WHICH SATISFIES BOUNDARY CONDITIONS
C
      Y1 = ABC
      IF (Y1.LT.0) PER = .1
C
C       IF(Y1.LT.0) PER=.9
C
      IF (Y1.GT.0) PER = 10.
C
C       IF(Y1.GT.0) PER=1.1
C
   80 CONTINUE
C      EITHER INCREASE OR DECREASE INITIAL GUESS TO COO
      COO = (PER)*COO
C     STOP IF COO EXCEEDS BOUNDS
      IF (COO.LT.CMIN) STOP 5432
      IF (COO.GT.CMAX) STOP 2345
      X2 = COO
      CALL INTEG
      Y2 = ABC
      IF ((Y1*Y2).LT.0) GO TO 90
      X1 = COO
      Y1 = Y2
      GO TO 80
   90 CONTINUE
C
C     W1 AND W2 ARE WEIGHTING CONSTANTS FOR INTERVAL HALVING
C
C     Y1 AND Y2 ARE OF OPPOSITE SIGN
C
```

```
      W3 = ABS(Y1)+ABS(Y2)
      W1 = ABS(Y1)/W3
      W2 = ABS(Y2)/W3
      COO = W2*X1+W1*X2
  100 CONTINUE
      CALL INTEG
      X3 = COO
      Y3 = ABC
      IF (ABS(Y3).LT.0.001) GO TO 20
      IF ((Y1*Y3).LT.0) GO TO 110
C
C       Y1 & Y3 ARE OF THE SAME SIGN
C
      X1 = X3
      Y1 = Y3
      W3 = ABS(Y1)+ABS(Y2)
      W1 = ABS(Y1)/W3
      W2 = ABS(Y2)/W3
      COO = W2*X1+W1*X2
      GO TO 100
  110 CONTINUE
C
C       Y1 & Y3 ARE OF OPPOSITE SIGN
C
      X2 = X3
      Y2 = Y3
      W3 = ABS(Y1)+ABS(Y2)
      W1 = ABS(Y1)/W3
      W2 = ABS(Y2)/W3
      COO = W2*X1+W1*X2
      GO TO 100
C
      STOP 1234
      END



      FUNCTION CHSI1  (Z)
C
C
      COMMON /BLK4/ CHSI10,CHSI20,ABARO,Z1BAR,LC
      COMMON /BLK8/ ZOL
      REAL K1,K2,K3,K4,K5,K6,K7,K8,Q3,Q4,Q5,LC
      IF (Z.LT.ABARO) GO TO 10
      IF (Z.LT.Z1BAR) GO TO 20
C
C     Z GREATER THAN Z1BAR
C
   10 CHSI1 = 0.0
      RETURN
   20 CONTINUE
      CHSI1 = CHSI10
      RETURN
      END
```

```
      FUNCTION CHSI2 (Z)
C
C
      COMMON /BLK4/ CHSI10,CHSI20,ABARO,Z1BAR,LC
      COMMON /BLK8/ ZOL
      REAL K1,K2,K3,K4,K5,K6,K7,K8,Q3,Q4,Q5,LC
      IF (Z.LT.ABARO) GO TO 10
      IF (Z.LT.Z1BAR) GO TO 20
C
C     Z GREATER THAN Z1BAR
C
   10 CHSI2 = 0.0
      RETURN
   20 CONTINUE
      CHSI2 = CHSI20
      RETURN
      END



      SUBROUTINE COEFFS
C
C       THIS SUBROUTINE DEFINES THE COEFFICIENTS IN THE DIFFERENTIAL
C       EQUATIONS TO BE SOLVED.
C     IMPLICIT REAL*8(A-H,K,L,O-Z)
C
      COMMON /BLK2/ K1,K2,K3,K4,K5,K6,K7,K8,C1,C2,C3,C4,Q1,Q2,A,AD,V1,V2
     1    ,GG
      COMMON /BLK3/ B,B2,B3,C,A00,B00,EPSNU,OMEGA,C5,C6,Q3,Q4,Q5
      COMMON /BLK4/ CHSI10,CHSI20,ABARO,Z1BAR,LC
      COMMON /BLK7/ ABC,C00,CO,OMEG1,P,R1,R2,TM
      COMMON /BLK8/ ZOL
      REAL K1,K2,K3,K4,K5,K6,K7,K8,Q3,Q4,Q5,LC
C
C     COEFFICIENTS IN THE DIFFERENTIAL EQUATIONS
C
      OMEGA = OMEG1
      ABARO = 0.0
C
C      ABARO= START OF ILLUMINATION
C       Z1BAR=2*ZOL = POINT ON AXIS WHERE ILLUMINATION STOPS
C
      CHSI10 = (3.04E-3)*CO
      CHSI20 = (3.38E-2)*CO
      Z1BAR = 2*ZOL
      EPSNU = 1.5E-19
C
C      WATTS*SEC
C
      A00 = 2.0E17
      B00 = .443
C
C
```

```
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C       (CHSI1)     HV + RI --> R + I*
C       (CHSI2)     HV + I2 --> I* + I
C
C         (K1)      I* + R --> RI
C         (K2)      I + R --> RI
C         (K3)      R + R --> R2
C         (K4)      R + RI --> R2 + I
C         (K5)      R + I2 --> RI + I
C         (K6)      R + RI --> R2 + I*
C         (K7)      I* + RI --> I2 + R
C         (K8)      I + RI --> I2 + R
C
C         (Q1)      I* + RI --> I + RI
C         (Q2)      I* + I2 --> I + I2
C         (Q3)      I* + R --> I + R
C         (Q4)      I* + R2 --> I + R2
C         (Q5)      I* + I --> I + I
C
C         (C1)      I* + I + RI --> I2 + R1
C         (C2)      I + I + RI --> I2 + RI
C         (C3)      I* + I + I2 --> I2 + I2
C         (C4)      I + I + I2 --> I2 + I2
C         (C5)      I + I + R2 --> I2 + R2
C         (C6)      I + I* + R2 --> I2 + R2
C
C
C       WALL REACTIONS
C
C        (V1)      I + I + WALL --> I2 + WALL
C        (V2)     R2 + I + WALL --> R + RI + WALL
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C     K1 = 34.7E-13
C     K1 = 5.6E-13
C
      K1 = .9E-13
C
      K2 = 8.05E-11
C
C     K2 = 2.3E-11
C     K2 = .657E-11
C
C     K3 = 10.4E-12
C     K3 = 2.6E-12
C
      K3 = .65E-12
C
C     K4 =     9.E-16
C     K4 = 3.E-16
C
      K4 = 1.E-16
C
```

```
      K5 = 3.E-11
C
C     K5 = 1.0E-11
C     K5 =     .33E-11
C
C     K6 = 10.24E-17
C     K6 = 3.2E-17
C
      K6 = 1.E-17
C
C     K7 = 4.5E-19
C     K7 = 3.0E-19
C
      K7 = 1.5E-19
C
C     K8 =      4.8E-23
C     K8 = 1.6E-23
C     K8 =      .533E-23
C
      K8 = 0.0
C
C     Q1 = 8.4E-16
C     Q1 = 2.0E-16
C
      Q1 = .476E-16
C
C     Q2 = 4.94E-11
C     Q2 = 1.9E-11
C
      Q2 = .73E-11
C
C     Q3 =      11.1E-18
C     Q3 = 3.7E-18
C
      Q3 = 1.23E-18
C
C     Q4 =     14.1E-16
C     Q4 = 4.7E-16
C
      Q4 = 1.57E-16
C
C     Q5 =     4.8E-14
C     Q5 = 1.6E-14
C
      Q5 = .53E-14
C
C     C1 = 10.2E-33
C     C1 = 3.2E-33
C
      C1 = 1.E-33
C
      C2 = 45.E-32
C
C     C2 = 8.5E-32
C     C2 = 1.6E-32
```

```
C
C      C3 = 14.4E-32
C      C3 = 8.E-32
C
       C3 = 4.44E-32
C
       C4 = 4.94E-30
C
C      C4 = 3.8E-30
C      C4 = 2.92E-30
C
C      C5 = 6.E-31
C      C5 = 4.8E-31
C
       C5 = 3.6E-31
C
C      C6 =      2.25E-32
C      C6 =      1.8E-32
C      C6 =      1.35E-32
C
       C6 = 0.0
C
C      V1 =      3.0E-12
C      V1 =      1.0E-12
C      V1 =      .33E-12
C
       V1 = 0.0
C
C      V2 =      3.0E-11
C      V2 =      1.0E-11
C      V2 =       .33E-11
C
       V2 = 0.0
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C      K1=.903E-13
C      K2=80.E-12
C      K3=.65E-12
C      K4=1.000E-16
C      K5=3.009E-11
C      K6=1.0E-17
C      K7=.1517E-18
C      K8=1.6E-23
C
C      Q1=.476E-16
C      Q2=1.9E-11
C      Q3=.1235E-17
C      Q4=1.57E-16
C      Q5=.53E-14
C
C      C1=1.053E-33
C      C2=45.0E-32
C      C3=.4447E-31
C      C4=4.94E-30
```

```
C       C5=3.6E-31
C       C6=       1.8E-32
C
C       V1=        1.0E-12
C       V2=        1.0E-11
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
        GG = 2*(.18/LC)**2
        C = 3.0E10
        B = P*(3.5E16)
        B2 = B*B
        B3 = B2*B
C
C         WRITE OUT COEFFICIENTS
C
        WRITE (6,10) K1,K7,Q1,C5
        WRITE (6,20) K2,K8,Q2,V1
        WRITE (6,30) K3,C1,Q3,V2
        WRITE (6,40) K4,C2,Q4
        WRITE (6,50) K5,C3,Q5
        WRITE (6,60) K6,C4,C6
   10 FORMAT (T5,'K1 = ',E15.7,T30,'K7 = ',E15.7,T60,'Q1 = ',E15.7,T85,
     1    'C5 = ',E15.7)
   20 FORMAT (T5,'K2 = ',E15.7,T30,'K8 = ',E15.7,T60,'Q2 = ',E15.7,T85,
     1    'V1 = ',E15.7)
   30 FORMAT (T5,'K3 = ',E15.7,T30,'C1 = ',E15.7,T60,'Q3 = ',E15.7,T85,
     1    'V2 = ',E15.7)
   40 FORMAT (T5,'K4 = ',E15.7,T30,'C2 = ',E15.7,T60,'Q4 = ',E15.7)
   50 FORMAT (T5,'K5 = ',E15.7,T30,'C3= ,',E15.7,T60,'Q5 = ',E15.7)
   60 FORMAT (T5,'K6 = ',E15.7,T30,'C4= ,',E15.7,T60,'C6 = ',E15.7)
        RETURN
        END
```

```
      SUBROUTINE FUN (N,Z,Y,F)
C        THIS SUBROUTINE DEFINES THE RIGHT HAND SIDE
C        OF THE DIFFERENTIAL EQUATIONS FOR THE CHEMICAL KINETICS
C     IMPLICIT REAL*8(A-H,K,L,O-Z)
      DIMENSION Y(7),F(7)
      COMMON /BLK1/ X7,POWER
      EXTERNAL CHSI1,CHSI2
      COMMON /BLK2/ K1,K2,K3,K4,K5,K6,K7,K8,C1,C2,C3,C4,Q1,Q2,A,AD,V1,V2
     1    ,GG
      COMMON /BLK3/ B,B2,B3,C,A00,B00,EPSNU,OMEGA,C5,C6,Q3,Q4,Q5
      COMMON /BLK4/ CHSI10,CHSI20,ABAR0,Z1BAR,LC
      COMMON /BLK7/ ABC,C00,C0,OMEG1,P,R1,R2,TM
      REAL K1,K2,K3,K4,K5,K6,K7,K8,Q3,Q4,Q5,LC
C     QY=QUANTUM YIELD
      QY = 1.0
C        F(I),I=1,6 ARE RATES OF CHANGES FOR THE CONCENTRATIONS
C          F(1)=D[RI]/DZ                F(2)=D[R]/DZ
C          F(3)=D[R2]/DZ                F(4)=D[I2]/DZ
C          F(5)=D[I*]/DZ                F(6)=D[I]/DZ
      X8 = C00/(Y(7)*B)
      X7STAR = Y(7)*B+X8
      DIF = Y(5)-.5*Y(6)
      CALL SIGMA (SIG2)
      SIG = SIG2
      F(1) = K1*B*Y(2)*Y(5)+K2*B*Y(2)*Y(6)-CHSI1(Z)*Y(1)-K4*B*Y(1)*Y(2)+
     1   K5*B*Y(2)*Y(4)-K7*B*Y(5)*Y(1)-K6*B*Y(2)*Y(1)+V2*B*Y(3)*Y(6)-K8*
     2   B*Y(6)*Y(1)
      F(2) = CHSI1(Z)*Y(1)-K1*B*Y(2)*Y(5)-K2*B*Y(2)*Y(6)-2*K3*B*Y(2)*Y(2
     1   )-K4*B*Y(1)*Y(2)-K6*B*Y(1)*Y(2)-K5*B*Y(2)*Y(4)+V2*B*Y(3)*Y(6)+
     2   K7*B*Y(5)*Y(1)+K8*B*Y(6)*Y(1)
      F(3) = K3*B*Y(2)*Y(2)+K6*B*Y(1)*Y(2)+K4*B*Y(1)*Y(2)-V2*B*Y(3)*Y(6)
      A1 = C1*B2*Y(1)*Y(5)*Y(6)+C2*B2*Y(1)*Y(6)*Y(6)+
     1    C3*B2*Y(4)*Y(5)*Y(6)
      A2 = C4*B2*Y(4)*Y(6)*Y(6)-CHSI2(Z)*Y(4)+K7*B*Y(5)*Y(1)-
     1   K5*B*Y(2)*Y(4)+V1*B*Y(6)*Y(6)+C5*B2*Y(6)*Y(6)*Y(3)
      F(4) = A1+A2+K8*B*Y(6)*Y(1)+C6*B2*Y(6)*Y(5)*Y(3)
      A3 = QY*CHSI1(Z)*Y(1)+CHSI2(Z)*Y(4)-K1*B*Y(2)*Y(5)
      A4 = -C1*B2*Y(1)*Y(5)*Y(6)-C3*B2*Y(4)*Y(5)*Y(6)-Q1*B*Y(1)*Y(5)
      A5 = -Q2*B*Y(4)*Y(5)-C*SIG*X7STAR*DIF+K6*B*Y(2)*Y(1)
      F(5) = A3+A4+A5-Q3*B*Y(5)*Y(2)-Q4*B*Y(5)*Y(3)-Q5*B*Y(5)*Y(6)-K7*B*
     1   Y(5)*Y(1)-C6*B2*Y(6)*Y(5)*Y(3)
      A6 = CHSI2(Z)*Y(4)+Q1*B*Y(1)*Y(5)+Q2*B*Y(4)*Y(5)-2*C5*B2*Y(6)*Y(6)
     1   *Y(3)-K8*B*Y(6)*Y(1)
      A7 = C*SIG*X7STAR*DIF-C1*B2*Y(1)*Y(5)*Y(6)
      A8 = -2*C2*B2*Y(1)*Y(6)*Y(6)-C3*B2*Y(4)*Y(5)*Y(6)
      A9 = -2*C4*B2*Y(4)*Y(6)*Y(6)-K2*B*Y(2)*Y(6)+K4*B*Y(1)*Y(2)
      A10 = Q3*B*Y(5)*Y(2)+Q4*B*Y(5)*Y(3)+Q5*B*Y(5)*Y(6)+K5*B*Y(2)*Y(4)-
     1   V2*B*Y(3)*Y(6)-2*V1*B*Y(6)*Y(6)
      F(6) = A6+A7+A8+A9+A10-C6*B2*Y(6)*Y(5)*Y(3)
      DO 10 I = 1, 6
         F(I) = F(I)/OMEGA
   10 CONTINUE
      F(7) = Y(7)*DIF*B*SIG
      RETURN
      END
```

```
      SUBROUTINE SIGMA (SIG)
C      THIS SUBROUTINE DEFINES THE CROSS SECTION SIGMA
C
      COMMON /BLK3/ B,B2,B3,C,A00,B00,EPSNU,OMEGA,C5,C6,Q3,Q4,Q5
      COMMON /BLK7/ ABC,C00,C0,OMEG1,P,R1,R2,TM
      REAL NU,NUS,NU0,NU1,NU2,NU3,NU4,NU5
      PI = 3.14159
      PIS = PI*PI
      NU = C/1.315246E-4
      NUS = NU*NU
      PISNUS = PIS*NUS*4.
      G = 0
      CS = C*C
      NU0 = NU
      NU1 = NU0+.141*C
      NU2 = NU1+.068*C
      NU3 = NU0-.427*C
      NU4 = NU3-.026*C
      NU5 = NU4-.068*C
      DELTA23 = NU-NU5
      DELTA22 = NU-NU4
      DELTA21 = NU-NU3
      DELTA34 = NU-NU0
      DELTA33 = NU-NU1
      DELTA32 = NU-NU2
      TEMPO = 293
      TWALL = TEMPO
      T1 = TWALL
      A = 5.434
      A1 = A*2.4/7.7*CS
      A2 = A*3.0/7.7*CS
      A3 = A*2.3/7.7*CS
      A4 = A*5.0/7.7*CS
      A5 = A*2.2/7.7*CS
      A6 = A*0.6/7.7*CS
      FUGTEMP = SQRT(T1/300.)
      ALPHAM = 1.88E7*FUGTEMP
      DELDOP = 2.51E8*FUGTEMP
      DELNU = DELDOP+ALPHAM*P
      SIGMA23 = A1/(PISNUS*DELNU)/(1+(2.*DELTA23/DELNU)**2)*5./12.
      SIGMA22 = A2/(PISNUS*DELNU)/(1+(2.*DELTA22/DELNU)**2)*5./12.
      SIGMA21 = A3/(PISNUS*DELNU)/(1+(2.*DELTA21/DELNU)**2)*5./12.
      SIGMA34 = A4/(PISNUS*DELNU)/(1+(2.*DELTA34/DELNU)**2)*7./12.
      SIGMA33 = A5/(PISNUS*DELNU)/(1+(2.*DELTA33/DELNU)**2)*7./12.
      SIGMA32 = A6/(PISNUS*DELNU)/(1+(2.*DELTA32/DELNU)**2)*7./12.
      SIGMAT = SIGMA23+SIGMA22+SIGMA21+SIGMA34+SIGMA33+SIGMA32
      SIG = SIGMAT
      RETURN
      END
```

```
      SUBROUTINE INTEG
C
C      THIS SUBROUTINE INTEGRATES THE SYSTEM OF DIFFERENTIAL EQUATIONS
C      USING A VARIABLE STEP SIZE 7TH ORDER RUNGE KUTTA METHOD.
C     IMPLICIT REAL*8(A-H,K,L,O-Z)
C
      DIMENSION Y(7),F(7),YO(7),X(7),WK(49)
      COMMON /BLK1/ X7,POWER
      COMMON /BLK2/ K1,K2,K3,K4,K5,K6,K7,K8,C1,C2,C3,C4,Q1,Q2,A,AD,V1,V2
     1    ,GG
      COMMON /BLK3/ B,B2,B3,C,A00,B00,EPSNU,OMEGA,C5,C6,Q3,Q4,Q5
      COMMON /BLK4/ CHSI10,CHSI20,ABARO,Z1BAR,LC
      COMMON /BLK7/ ABC,C00,C0,OMEG1,P,R1,R2,TM
      EXTERNAL FUN,CHSI1,CHSI2
      REAL K1,K2,K3,K4,K5,K6,K7,K8,Q3,Q4,Q5,LC
C
C     INTEGRATE SYSTEM FROM Z=0 TO Z=LC USING RUNGE-KUTTA METHOD
C
C     X(1)=RI
C     X(2)=R
C     X(3)=R2
C     X(4)=I2
C     X(5)=I*
C     X(6)=I
C     X(7)=RHO+
C     X8=RHO-
C     X9=I*-.5*I
C
      WW2 = (.98)**2
      N = 7
      TOL = 1.0E-6
      PD = 1.0
      MTH = 1
      H = 1.00
      HMIN = H/10000000.
      HMAX = H/100.
      HUSE = HMIN*10
      IERR = 0
C
C     INITIAL CONDITIONS
C
      Z0 = 0.0
      YO(1) = 1.0
      Z1 = 0.0
      DO 10 I = 2, 6
         YO(I) = 0.0
   10 CONTINUE
C
C       GUESS AT INITIAL CONDITIONS FOR X(7) AND X(8)
C        INITIAL CONDITION FOR X70=RHO+ DEPENDS UPON C00 VALUE
C
      X70 = SQRT(R1*C00)
      YO(7) = X70/B
C
C
```

```
C     PRINT HEADER
      WRITE (6,20)
   20 FORMAT (///,T7,'Z',T20,'[RI]',T32,'[R],',T45,'[R2],',T57,'[I2]',
     1   T69,'[I*]',T80,'[I] ',T91,'[RHO+]',T103,'[RHO-]',T112,
     2   'INVERSION')
   30 CONTINUE
C
C
      DO 40 I = 1, 7
         X(I) = B*YO(I)
   40 CONTINUE
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C       USE CONSERVATION LAWS---THE ABOVE SYSTEM OF DIFFERENTIAL
C       EQUATIONS HAS THE TWO INTEGRALS
C
C          [RI] + [R] +2[R2] = CONSTANT= B
C      OR
C           Y(1)  + Y(2)  + 2Y(3)= 1.0
C
C    AND    [RI]  + 2[I2] + [I*] + [I] = B
C      OR
C            Y(1) + 2Y(4) + Y(5) + Y(6) =1.0
C
      YOTH2 = 1.0-YO(1)-2.*YO(3)
      YOTH6 = 1.0-YO(5)-2.0*YO(4)-YO(1)
C      CHECK ON INTEGRATION ROUTINE COMPARED TO ABOVE INTEGRALS
C      TEST TO SEE IF YOTH2 IS ZERO
C
      ERROR1 = 0.0
      IF (YOTH2.EQ.0.0) GO TO 50
      ERROR1 = (YO(2)-YOTH2)*100/YOTH2
   50 CONTINUE
      ERROR2 = 0.0
C
C      TEST TO SEE IF YOTH6 IS ZERO
C
      IF (YOTH6.EQ.0.0) GO TO 60
      ERROR2 = (YO(6)-YOTH6)*100/YOTH6
   60 CONTINUE
      IF ((ERROR1.OR.ERROR2).GT..5) WRITE (6,70)
   70 FORMAT (1X,' ERROR1 OR ERROR2 IS OUT OF BOUNDS       ')
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
C   END OF TEST FOR CONSERVATION LAWS BEING SATISFIED
C
CCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCCC
C
      X8 = COO/X(7)
      X9 = X(5)-.5*X(6)
      X7STAR = X(7)+X8
C
C
```

```
C         USE SUBROUTINE SIGMA TO CALCULATE CROSS SECTION SIGMA
      CALL SIGMA (SIG2)
C     PRINT OUT VALUES FOR CONCENTRATIONS , RHO+, RHO- AND INVERSION
C
      WRITE (6,80) Z0,(X(I),I=1,7),X8,X9
   80 FORMAT (1X,E12.5,8E12.5,E12.5,E12.5)
C
C         USE 7TH ORDER RUNGE KUTTA INTEGRATION SCHEME WITH VARIABLE STEP
C         STEP SIZE CAN VARY FROM HMIN TO HMAX
C         Z0 IS STARTING VALUE FOR Z
C         Z1 IS NEXT STOPPING POINT IN INTEGRATION SCHEME
C         TOL IS TOLERENCE
C         IERR IS ERROR CODE TO DETERMINE IF INTEGRATION WAS SUCCESSFUL
C
      Z1 = Z1+H
      CALL RKF7 (N,Z0,Z1,Y0,TOL,FUN,PD,MTH,HMIN,HMAX,HUSE,WK,IERR)
      IF (IERR.NE.0) WRITE (6,90)
      IF (IERR.NE.0) STOP 1717
   90 FORMAT (1X,' IERR IS NOT ZERO                    ')
      X(7) = B*Y0(7)
      X8 = C00/X(7)
      IF ((Z1+.5*H).GE.(LC+.2)) GO TO 110
C
      GO TO 30
      DO 100 I = 1, 7
         X(I) = B*Y0(I)
  100 CONTINUE
      X8 = C00/X(7)
      X9 = X(5)-.5*X(6)
  110 CONTINUE
      XX7L = B*Y0(7)
      RHOPL = X70/(SQRT(R2*R1))
C
C      RHOPL=RHO-PLUS AT Z=L THEORETICAL VALUE
C      XX7L= CALCULATED VALUE OF RHO-PLUS AT Z=
C      ABC =DIFFERENCE=XX7L-RHOPL
C
      DIF = (((XX7L-RHOPL)/RHOPL)*100)
      ABC = DIF
      WRITE (6,120) DIF,RHOPL,XX7L,C00
  120 FORMAT (1X,'DIFFERENCE = ',E18.9,2X,'THEORETICAL=',E18.9,2X,
     1    ' ACTUAL = ',E18.9,2X,'C00 = ',E18.8)
C
      XX7L = X(7)
      TM = 1-R2/(WW2)
      IF (TM.LT.0) GO TO 30
      POWER = EPSNU*TM*C*XX7L
      WRITE (6,130) R1,R2,POWER,TM,Z1,P
  130 FORMAT (1X,'R1 = ',F10.7,1X,'R2 = ',F10.7,1X,'POWER =',E18.10,1X,
     1    'TM = ',F10.8,1X,'L = ',F15.7,'P = ',F15.7)
      IF ((Z1+.5*H).GE.(LC+.2)) GO TO 140
      GO TO 30
  140 CONTINUE
      RETURN
      END
```